CBESSK

PURPOSE

Compute the real or complex component of the modified Bessel function of the third kind and order v for a complex argument where v is a non-negative real number.

DESCRIPTION

The modified Bessel function of the third kind can be defined in terms of the modified Bessel function of the first kind:

$$K_{\nu}(z) = \frac{\pi}{2}\left(\frac{I_{-\nu}(z) - I_{\nu}(z)}{\sin(\pi\nu)}\right) \qquad \textbf{(EQ Aux-59)}$$

where z is a complex number and I_v is the modified Bessel function of the first kind. See the documentation for the CBESSI command for details on this function.

The real part of the input argument must be less than the logarithm of the largest single precision number on the given computer. The order is restricted to values between 0 and 100.

SYNTAX 1

LET <y2> = CBESSKR(<r1>,<i1>,<v>) <SUBSET/EXCEPT/FOR qualification>

where <r1> is the real component of a number, variable or parameter;
<i1> is the complex component of a number, variable or parameter;
<v> is a non-negative number, variable, or parameter that specifies the order of the Bessel function;
<y2> is a variable or a parameter (depending on what <r1> and <i1> are) where the computed Bessel value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the real component.

SYNTAX 2

LET <y2> = CBESSKI(<r1>,<i1>,<v>) <SUBSET/EXCEPT/FOR qualification>

where <r1> is the real component of a number, variable or parameter;
<i1> is the complex component of a number, variable or parameter;
<v> is a non-negative number, variable, or parameter that specifies the order of the Bessel function;
<y2> is a variable or a parameter (depending on what <r1> and <i1> are) where the computed Bessel value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax computes the complex component.

EXAMPLES

LET XR = CBESSKR(2,1,2)
LET XC = CBESSKI(2,1,2)
LET AR = CBESSKR(R1,C1,3)
LET AC = CBESSKI(R1,C1,3)

NOTE 1

DATAPLOT uses the routine BESKCF from the BESPAK library. This library was written by David Sagin (Sookne), Computer Center, Tel Aviv University.

NOTE 2

DATAPLOT does not calculate this function for negative orders. However, the following relation can be used:

$$K_{-\nu}(z) = K_{\nu}(z) \qquad \textbf{(EQ Aux-60)}$$

where z is a complex number.

NOTE 3

The Kelvin functions of the second kind are defined as follows:

$$\mathrm{ker}_{\nu}(x) + i\ \mathrm{kei}_{\nu}(x) = e^{\frac{-\nu\pi}{2}}\ \mathrm{K}_{\nu}\left(xe^{\frac{\pi i}{4}}\right) \quad \textbf{(EQ Aux-61)}$$

The CEXP and CEXPI functions can be used to compute the arguments to the CBESSKR and CBESSKI functions. The PROGRAM example below demonstrates this for Kelvin functions of order 0 and 1.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

BESSKN	=	Compute the modified Bessel function of the third kind, order N and real argument.
CBESSIR	=	Compute the real component of the modified Bessel function of order N and complex argument.
CBESSII	=	Compute the complex component of the modified Bessel function of order N and complex argument.
CBESSJR	=	Compute the real component of the Bessel function of the first kind, order N, and complex argument.
CBESSJI	=	Compute the complex component of the Bessel function of the first kind, order N, and complex argument.
CBESSYR	=	Compute the real component of the Bessel function of the second kind, order N, and complex argument.
CBESSYI	=	Compute the complex component of the Bessel function of the second kind, order N, and complex argument.

REFERENCE

“Handbook of Mathematical Functions, Applied Mathematics Series, Vol. 55,” Abramowitz and Stegun, National Bureau of Standards, 1964 (pages 355-433).

“Note on Backward Recurrence Algorithms,” Olver and Sookne, Mathematics of Computation, Volume 26, October 1972.

“Recurrence Techniques for the Calculation of Bessel Functions,” Goldstein and Thaler, Mathematics of Computation, Volume 13, April 1959.

“Bessel Functions of Complex Argument and Integer Order,” Sookne, Journal of Research of the National Bureau of Standards, Series B, Volume 77A, July-December, 1973.

APPLICATIONS

Special Functions

IMPLEMENTATION DATE

94/10

PROGRAM

```
. Kelvin functions of the second kind
DIMENSION 20 COLUMNS
LET X = SEQUENCE 0.2 0.2 10; LET N = SIZE X
LET XI = 0 FOR I = 1 1 N
LET ORDER = 0; LET CONST1 = PI/4
LET AR = CEXP(0,CONST1)
LET AI = CEXPI(0,CONST1)
LET CONST2 = -PI*ORDER/2
LET BR = CEXP(0,CONST2)
LET BI = CEXPI(0,CONST2)
LET CR = AR FOR I = 1 1 N
LET CI = AI FOR I = 1 1 N
LET DR = BR FOR I = 1 1 N
LET DI = BI FOR I = 1 1 N
LET YR YC = COMPLEX MULTIPLICATION X XI CR CI
LET KER0 = CBESSKR(YR,YC,ORDER)
LET KEI0 = CBESSKI(YR,YC,ORDER)
LET KER0 KEI0 = COMPLEX MULTIPLICATION DR DI KER0 KEI0
LET ORDER = 1; LET CONST2 = -PI*ORDER/2
LET BR = CEXP(0,CONST2)
LET BI = CEXPI(0,CONST2)
LET DR = BR FOR I = 1 1 N
LET DI = BI FOR I = 1 1 N
LET KER1 = CBESSKR(YR,YC,ORDER)
LET KEI1 = CBESSKI(YR,YC,ORDER)
LET KER1 KEI1 = COMPLEX MULTIPLICATION DR DI KER1 KEI1
YLIMITS -0.5 0.5; LINE SOLID DASH DOT DOT
TITLE KELVIN FUNCTIONS OF ORDER 0 AND 1
PLOT KER0 KEI0 KER1 KEI1 VS X
```